Mechanically verifying the correctness
of an offline partial evaluator *

John Hatcliff T
DIKU
Computer Science Department
Copenhagen University ¥

September 10, 1995

Abstract

We show that using deductive systems to specify an offline partial evaluator allows its correctness to be
mechanically verified.

For a A-mix-style partial evaluator, we specify binding-time constraints using a natural-deduction logic,
and the associated program specializer using natural (aka “deductive”) semantics. These deductive systems
can be directly encoded in the Elf programming language — a logic programming language based on the
LF logical framework. The specifications are then executable as logic programs. This provides a prototype
implementation of the partial evaluator.

Moreover, since deductive system proofs are accessible as objects in Elf, many aspects of the partial evalu-
ation correctness proofs (e.g., the correctness of binding-time analysis) can be coded in Elf and mechanically
verified.

This work illustrates the utility of declarative programming and of using deductive systems for defining
program specialization systems: by exploiting the logical character of definitions, one may specify, prototype,
and mechanically verify correctness via meta-programming — all within a single framework.

1 Introduction

As the application of program-specialization techniques increases, it 1s important to study ways of formally
specifying and proving these techniques correct. Unfortunately, relatively few specialization systems have been
proven correct, and we are not aware of any that have been mechanically verified. Even when systems have
been proven correct, it is unclear from the literature which styles of formalization are preferable.

As an example, consider offline partial evaluation of functional programs [5, 20]. This consists of two
phases: a binding-time analysis phase (where information is gathered about what parts of the source program
depend on known or unknown data), and a specialization phase (where constructs depending on known data are
reduced away). Denotational semantics has been the most widely applied formalism for specifying offline partial
evaluators. The denotational method allows both phases to be specified in the same framework: the analysis
phase is specified as an abstract interpretation [6, 21] and the specialization phase is specified as a function
from a source program to a residual program. Moreover, a prototype implementation can often be obtained
by a fairly straightforward (though informal) transliteration of the denotational specifications into functional
programs.

Recent work specifies the analysis phase using type systems while the specialization phase is specified deno-
tationally [10, 22]. Unfortunately, this approach loses a degree of synergy since both phases are not specified in
the same framework. For example, one can no longer obtain a prototype wia a simple translation to functional
programs (since the typing rules which specify the analysis cannot be simply translated). Others define the

program specializer using operational semantics and thus avoid what Wand has characterized as a “morass of
model-theoretic details” [37] associated with the denotational framework.

Note that the type system and operational semantics formalisms can be unified if one emphasizes their logical
character: a type-based analysis is a logic for deducing program properties, and an operational semantics is a
logic for deducing computational steps or input/output behaviour of programs. However, in program special-
ization systems that use these formalisms, this logical character has neither been emphasized nor exploited.

In this paper, we exploit the logical character of such deductive systems and obtain a uniform framework for
specifying, prototyping, and mechanically verifying the correctness of program specialization systems. Specif-
ically, we consider offline partial evaluation in the style of the partial evaluator A-miz [11].} Ad-mix is a good
illustrative case since it 1s simple, and one of the few partial evaluators with a rigorous semantic foundation. Tt
has also spawned additional work on the correctness of binding-time analysis and specialization [27, 37].

The results of the present work are as follows.

e We give specifications of binding-time constraints and specialization as natural-deduction style logics.
These specifications simplify meta-theory activities such as proving the correctness of binding-time analysis
and specialization.

e We formalize the specifications using LF — a meta-language (a dependently-typed A-calculus) for defining
logics [19]. In LF, judgements (assertions) are represented as types, and deductions are represented as
objects. Determining the validity of a deduction is reduced to checking if the representing object is
well-typed. Since LF type-checking is decidable, purported deductions can be checked automatically for
validity.

o We obtain prototypes directly from the formal specifications using EIf — a logic programming language
based on LF [29]. EIf gives an operational interpretation to LF types by treating them as goals. Thus,
the LF specifications of the binding-time analysis and specializer are directly executable in EIf.

e We formalize and mechanically verify much of the meta-theory of offline partial evaluation (e.g., correct-
ness of binding-time analysis and soundness of the specializer) via meta-programming in Elf. Correctness
conditions are formalized as judgements about “lower-level” deductions describing object-language evalua-
tion and transformation. Proofs of correctness are formalized as deductions of the correctness judgements.
EIf type-checking mechanically verifies that these deductions (and hence the correctness proofs) are valid.?

This methodology of specification /implementation /verification using LF and Elf has been successfully applied
in other problem areas [18, 23]. In particular, we build on Hannan and Pfenning’s work on compiler verification
in EIf [18]. They conjectured that their techniques could also be applied to partial evaluation [18, p. 416]. They
also identify the verification of transformations based on flow analyses as a “challenging problem, yet to be
addressed” [18, p. 415]. The present work addresses both of these points. We confirm their conjecture that LF
and EIf can be used for specification/implementation/verification of partial evaluators. Moreover, we give one
instance where transformations based on flow analyses can be verified — namely, the specialization of programs
based on binding-time analysis. Hannan has recently reported on a second instance: the use of EIf to encode a
type system for closure conversion [16].

The rest of the paper is organized as follows. Section 2 summarizes LF and EIf. Section 3 presents the
object language and its encoding in Elf. Section 4 presents the specifications of the binding-time analysis and
specializer. Section 5 illustrates how these specifications are executable in Elf. Section 6 shows how meta-
theoretic properties of the specifications can be mechanically verified. Section 7 surveys related work, and
Section 8 concludes.

2 LF and the EIf Programming Language

This section presents a very brief overview of LF and ELf. The reader is referred to the original work on LF by
Harper, Honsell, Plotkin [19] for a detailed presentation of LF. Pfenning [29] gives the technical foundations of

1 Our setting differs from that of A-mix in two ways: (1) we are not concerned with self-applying the partial evaluator, and (2)
our object language is typed while A-mix’s is untyped. However, the techniques here apply equally well to the untyped object
language of A-mix (in fact, they are simpler in the untyped case).

2This paper contains all the required EIf code and is implemented using version 0.3 of EIf. The code is available upon request
from the author.



Elf. Papers by Hannan and Pfenning [18] and Michaylov and Pfenning [23] give applications of Elf that provide
the main techniques used in the present work.
2.1 LF — a framework for defining logics

The LF calculus has three levels: objects, families, and kinds. Families are classified by kinds, and objects are
classified by types, i.e., families of kind Type.

Kinds K == Type | llz:A. K
Families A == a | lMaz:A1. Ay | A Ay As | AM
Objects M == c | x | Ae:A. M | My M

Family-level constants are denoted by a, and object-level constants by ¢. A; — Ay abbreviates Ilz: A;. A5 and
Iz : A. K when z does not appear free in A3 or K, respectively. The typing rules for LF can be found in [19].
We take 1-equivalence as the notion of definitional equality in LF [19, Appendix A.3]. For all the languages
we consider, we identify terms up to renaming of bound variables.

One defines a logic in LF by specifying a signature which declares the kinds of family-level constants a
and types of object-level constants ¢. These constants are constructors for the logic’s syntax, judgements,
and deductions. Well-formedness is enforced by LF type-checking. The LF type system can represent the
conditions associated with binding operators, with schematic abstraction and instantiation, and with the variable
occurrence and discharge conditions associated with rules in systems of natural deduction.

2.2 EIlf — an implementation of LF

The syntax of Elf is as follows (the last column lists the corresponding LF term). Optional components are
enclosed in (-).

kindexp 1= type Type
| {ud(:famezp)} kindexp Mz:A. K
| famexp => kindezp A—- K
famezp = id a
| {id(:famezp,)} famezp, IMa: Aq. Ay
| Lid(:famezp,)] famezp, Az A1, As
| famezp objexp AM
| famezpy => famezp, A — Ay
| famezp, <~ famezp, Ay — Ay
objexp = id c
| [id(:famezp)] objexp Ax: A M
| objexp, objexp, My M,

The terminal id ranges over variables, and family and object constants. Bound variables and constants in EIf can
be arbitrary identifiers, but free variables in a declaration or query must begin with an upper case letter. Free
variables act as logic variables and are implicitly TI-abstracted. Elf’s term reconstruction phase (preprocessing)
inserts these abstractions as well as appropriate arguments to these abstractions. Tt also fills in the omitted
types in quantifications {z} and abstractions [#] and omitted types or objects indicated by an underscore _.
The <= is used to improve the readability of some EIf programs. B <- A is parsed into the same representation
as A =>B; -> is right associative, while <- 1is left associative. An Elf program is a representation of an LF

signature. Although we are implicitly encoding logics in LF, we give all encodings using the syntax of EIf.

3 The Object Language

3.1 Syntax



Object language:
e € FExp T € Typ
e u= x| 0]|sce | pre | if0eieses T u= nat | 71 o™

lamz .e | appeger | fixz.e

Elf encoding:
exp : type. typ : type.
Z . exp. nat : typ.
SC : exp —> exp. arrow : typ —> typ —-> typ.

Pr : exp —> exp.

if : exp -> exp -> exp -> exp.
lam : (exp -> exp) -> exp.
app : exp —> exp -—> exp.

fix : (exp -> exp) -> exp.

Figure 1: The object language A

Figure 1 presents the syntax of the object language A (a PCF-like language) and its encoding in Elf. We
identify object-level terms up to a-equivalence (i.e., up to renaming of bound variables). We write e; = es
when e; and ey are a-equivalent.

The Elf signature for the object language syntax includes family constants exp and typ, and object constants
for each term and type constructor. Variable binding in lam 2 . ¢ and fix z. ¢ is represented using binding in the
meta-language (i.e., using higher-order abstract syntax [31]) — an idea going back to Church [3]. For example,
the expression

app (lamzy .lamzs . app (lam zs . 23) 1) 0

is encoded as
app (lam [x1] lam [x2] app (1am [x3] x3) x1) z.

In this style of encoding, the body of the lam expression
lamzy . lamzy . app (lam 23 . 23) 21

is represented as a “higher-order term” that maps expressions to expressions. This allows substitution in the
object language to be represented using Elf application and f-reduction. For example,

(lamzo.app (lamzs. 23) z1)[z1 = 2]

is represented in Elf as
([x1] 1am [x2] app (lam [x3] x3) x1) z.

An Elf g-reduction gives the expected result of the substitution.
lam [x2] app (lam [x3] x3) z

This representation properly implements capture-free object-level substitution since it is expressed wia EIf
substitution (which the EIf implementation guarantees to be capture-free). Note that since variables in the
object language are identified with variables in the meta-language, there is no explicit representation of identifiers
in the Elf signature for the object language.

Technically, one must ensure that encodings of syntax into Elf are adequate, i.e., that there is a compositional
bijection between the syntactic entities in the logical system and well-formed LF long #n-normal forms under the
given signature. An adequate encoding ensures that each entity is encoded uniquely and that no representations
of additional entities are introduced. All the encodings we use are adequate. See Harper et al.[19] for a detailed
discussion and proofs of adequacy for encodings similar to the ones used here.



tp_z: tp 0 : nat

tp_sc: tp e : nat
tp sce : nat

tp_pr: tp e : nat
tp pre : nat

tp_end : tp e : nat tpey: T tpes: T
tpifei eses: T

) (tpx:m)
tp_lam : Ipe:m
tplamz.e:m — 1
tp_app : tpeg:m — T2 tp e mwipr
tp appeper 1 T
tp_fiz (tpx:T)

tpe: T
tpfixz.e: T

Figure 2: The typing rules for A

3.2 Typing
3.2.1 Natural deduction style rules

Figure 3 presents the typing rules for A in a natural-deduction style logic (we use the notation of Prawitz
[33]).% Parentheses in the hypotheses of the rules tp_lam and tp_fir indicate the discharging of zero or more
occurrences of assumptions. We write T F tp e : 7 when #p e : 7 is derivable under undischarged assumptions
T. We shall only be concerned with assumptions involving identifiers (e.g., tp  : 7) since the logic only
allows discharging of assumptions of this form. For simplicity, we require identifiers z; in assumption sets
T = {tpx1:7, ..., tp Ty : Tn}, to be pairwise distinct.

3.2.2 Elf encoding

Figure 3 presents the Elf encoding of the typing rules. The figure begins with a declaration of the typing
judgement tp as a relation between source expressions exp and types typ. The implicit universal quantification
of the upper case variables captures the schematic nature of the rules. The rules for binding constructs lam
and fix (which involve hypothetical proofs in the premises) are encoded in Elf as proof constructors that take
proofs of hypothetical judgements as arguments. Such proofs are represented as functions mapping proofs of
assumption judgements to proofs of consequent judgements. The higher-order syntax representation requires
that judgements involving identifiers be represented as schematic judgements (i.e., identifiers are TI-quantified).
This is the case in rules tp_lam and tp_fix where e.g., the judgement tp (E x) T2 expresses that E may be
instantiated to representations of terms with free occurrences of z [19, Section 3.1].

3See the textbook of Tennent [36] for extensive examples using natural deduction logics to specify syntax and typing.



%%% typing judgement

tp : exp -> typ —> type.
%%% typing axioms and rules
tp_z : tp z nat.

tp_sc : tp (sc E) nat
<- tp E nat.

tp_pr : tp (pr E) nat
<- tp E nat.

tp_if : tp (if E1 E2 E3) T
<- tp El1 nat
<-tpE2T
<- tp E3 T.

tp_lam : tp (lam E) (arrow T1 T2)
<- {x:exp} tp x T1 -> tp (E x) T2.

tp_app : tp (app EO E1) T2
<- tp EO (arrow T1 T2)
<- tp E1 T1.

tp_fix : tp (fix E) (arrow T1 T2)
<- {x:exp} tp x (arrow T1 T2) -> tp (E x) (arrow T1 T2).

Figure 3: The Elf encoding of the typing rules




nt_z : 0 Vint O

int_sc : e Jint v
sce int scv

int_prz : e Yint 0
pre Yint 0
mt_prsc : € Yint scv
pre Yint v

int_ifz : er Jine 0 ez Yint v

if0ey eg ez Yint v

int_ifsc: ey Jint scvq ez Yint v
if0 ey es ez Yint v

int_lam : lamy.e Yint lamy.e

int_app © eq VYint lamy . ef eoly = e1] Yint v
appeg ey it v

int_fir ely = fixy.e] VYine v
fixy. e Yint v

Figure 4: The natural semantics rules for A (interpretation logic)

3.2.3 Executing the specification

The Elf encoding of the typing rules (see Figure 3) is an executable logic program. For example, type inference
for the example expression of Section 3.1 can be performed via the following EIf query.

?7- tp (app (lam [x1] lam [x2] app (lam [x3] x3) x1) z) T.
Solving...

T = arrow T1 nat.
5
no more solutions

The solution T = arrow T1 nat indicates that the example expression is a function yielding results of type
nat for any argument type T1.

3.3 Operational Semantics
3.3.1 Natural deduction style rules

Figure 4 presents the call-by-name natural semantics rules (i.e., the interpretation or evaluation logic) for A
terms. These rules correspond to the usual call-by-name “natural” or “deductive” operational semantics for
PCF [14, Chapter 4]. We write F e {};n: @ when e {;n; a is derivable, and e {};,: when there does not exist an
a such that F e {};n: a. It is easy to check that the relation induced by {};,: is a partial function.

Two terms are considered operationally equivalent if they cannot be distinguished by any program context

[14, Chapter 6].

-1



%h% interpreter judgement

int : exp -> exp -> type.

%%#% interpreter axioms and rules
int_z : int z z.

int_sc : int (sc E) (sc A)
<- int E A.

int_prz : int (pr E) z
<- int E z.

int_prsc : int (pr E) A
<- int E (sc A).

int_ifz : int (if E1 E2 E3) A
<- int E1 z
<- int E2 A.

int_ifsc : int (if E1 E2 E3) A
<- int E1 (sc A1)
<- int E3 A.
int_lam : int (lam E) (lam E).
int_app : int (app EO E1) A
<- int EO0 (lam EO’)
<- int (EO’ E1) A.
int_fix : int (fix E) A

<- int (E (fix E)) A.

Figure 5: The Elf encoding of the natural semantics rules (interpreter)

Definition 1 (Operational Equivalence) e1,e3 € A are operationally equivalent (written e; & ey if for all
conterts C' such that & tp eq : nat and F tp ey : nat one of the following holds:

1. e1 fint and es fline, or
2. Fer Yint a1 and F ey Yint az and a1 = as.
Section 6 shows that our partial evaluator produces an output program that is operationally equivalent to the

given input program.

3.3.2 EIf encoding

Figure 5 gives the Elf encoding of the natural semantics. The encoding techniques are similar to those used by
Michaylov and Pfenning [23]. In the rules int_app and int _fix, we take advantage of the higher-order abstract
syntax representation and use EIf application (i.e., f-reduction) to implement capture-free substitution.



Value rules: Elf encoding:

val_z val 0 %h% value judgement
val v val : exp —-> type.
val_sc: ————
val scv

%h% value logic

val_lam : wval lamz.e
val_z : val z.

val_sc : val (sc E)
<- val E.

val_lam : val (lam E).

Figure 6: The value rules and Elf encoding

3.3.3 Executing the specification

As with the typing rules, the EIf encoding of the natural semantics rules (see Figure 5) is an executable logic
program. For example, evaluation of the example expression of Section 3.1 can be performed via the following
Elf query.

?- int (app (lam [x1] lam [x2] app (lam [x3] x3) x1) z) A.

A = lam [x2:exp] app (lam [x3:exp] x3) z.

no more solutions

3.3.4 Canonical terms of evaluation

An examination of the rules of Figure 4 reveals that the canonical terms (i.e., values) of evaluation are either
0, sc e where ¢ is also canonical, and lamz .e’. We formalize the notion of canonical term with the judgement
val of Figure 6. The encoding of the value rules in Elf is straightforward.

The following proposition formalizes the above claim: terms satisfying the value judgement val are indeed
the results of evaluation.

Proposition 1 If Fe Y €, then F val €

The proof of this proposition proceeds by induction over derivations of e {};,; €’; it can be formalized and
mechanically checked in Elf. The reader is referred to the work of Michaylov and Pfenning [23] where a detailed
presentation of such a proof for call-by-value evaluation is given.

Finally, note that we have followed convention [14] and defined the value judgement on the syntactic category
of expressions while omitting any notion of typing. Thus, the judgement does not enforce well-typedness and
admits terms that would not result from the evaluation of a well-typed program. For example, F val sc (lamz . z)
but there does not exist a 7 such that F tpsc (lamz.z): 7. Well-typedness of values is expressed by using the
typing judgement in conjunction with the value judgement.

4 Specifying an Offline Partial Evaluator for A

A partial evaluator takes a source program p and a subset s of p’s input, and produces a residual program p;
which 1is specialized with respect to s. The correctness of the partial evaluator implies that running ps on p’s
remaining input d gives the same result as running p on the complete input s and d. The data s and d are



Annotated object language:

w €  Sexp m € Sder
w = y | Oy | scmw | pr, w | if0,, wi wyws | m = s |d
lam,, y.w | app,, wow | fix, y.w |
lift w
Elf encoding:
sexp : type. sder : type.
bz : sder -> sexp. s : sder.
bsc : sder -> sexp —-> sexp. d : sder.

bpr : sder -> sexp —-> sexp.

bif : sder -> sexp -> sexp -> sexp -> sexp.
blam : sder -> (sexp -> sexp) -> sexp.
bapp : sder -> sexp -> sexp -> sexp.

bfix : sder -> (sexp -> sexp) -> sexp.

lift : sexp -> sexp.

Figure 7: The annotated object language Ay,

often referred to as static and dynamic data (respectively) since s is fixed at specialization time whereas one
may supply various data d during runs of p;.

The specialized program p; is obtained from p by evaluating constructs that depend only on s, while rebuild-
ing constructs that may depend on dynamic data. Offline partial evaluation accomplishes this in two phases:
(1) a binding-time analysis phase, and (2) a specialization phase.

1. Binding-time analysis: Given assumptions about which program inputs are static and dynamic,
binding-time analysis constructs an annotated program where each program construct is annotated with
a specialization directive and a specialization type.

¢ Specialization directives: A construct is assigned a directive of eliminable if it depends only on
static data and thus can be completely evaluated during the specialization phase. A construct is
assigned a directive of residual if it may depend on dynamic data and thus must be reconstructed in
the specialization phase.

e Specialization types: The specialization types (a.k.a binding times) are the carriers of information
during the analysis phase. They describe the “knownness” or the “unknownness” of expressions. This
information is used to determine the specialization directives assigned to constructs. For example, if
the argument of a destructor construct has a specialization type indicating that it is unknown, then
that construct cannot be evaluated at specialization time and must be given a residual directive.

2. Specialization: During the specialization phase, the specializer simply follows the directives assigned dur-
ing binding-time analysis: eliminable constructs are evaluated (and thus eliminated); residual constructs
are reconstructed (and thus appear in the residual program).

Several different formalizations of binding-time analysis and specialization are given in the literature (see
the textbook of Jones et al. [20] for a survey). In the remainder of this section, we give a logical formalization
that is suitable for encoding in EIf.

4.1 Binding-time analysis

We first outline how binding-time information is expressed in a program annotated with specialization direc-
tives and types. Following this, we give a binding-time logic that determines which annotations are appropriate
for a given soure language term.

10



Specialization types: Elf encoding:

Prat € Styp[nat] styp :typ->type.
Prar 5= sta | dynog sta : stypnat.
dyn :stypT.

Primsm € Styp[ri — 7]

barrow : styp T1 ->styp T2 -> styp (arrow T1 T2).
Pri—7a = P P | dyn71—>7'2

Figure 8: The specialization types for Ay

4.1.1 Specialization directives

A binding-time analysis for A associates each A term with a term in the annotated language Ap; of Figure 7. An
annotated term is indexed by a specialization directive s or d indicating if it is eliminable (z.e., it depends only
on static data and thus can be eliminated at specialization time) or residual (i.e., it may depend on dynamic
data and thus must be placed in the residual program). Identifiers are not indexed since the appropriate
information can be determined from the environment. A coercion construct lift is added to Ay to residualize
the result of evaluating an eliminable term. This allows static computation to occur in a residual context (an
example is given in Section 4.3). A term w € Ay, is completely residual if it consists of only d-indexed constructs
and identifiers. For example, the following well-annotated version of the example expression of Section 3.1 is
completely residual.

app, (lamg yy . lamg ys . app, (lamg ys . y3) 1) 04

Intuitively, the specializer will output completely residual terms — all eliminable constructs will have been
evaluated (this will be proven in Section 6.1). The EIf encoding of Ay terms follows that of A terms (except
that directive indexing is captured by supplying an extra argument of type sder to a constructor). For example,
the term above is encoded as follows.

bapp d (blam d [y1] (blam d [y2] (bapp d (blam d [y3] y3) y1))) (bz d)

4.1.2 Specialization types

Figure 8 presents a m-indexed family of specialization types for Ay (we omit type indices on specialization types
when they can be inferred from the context). A specialization type ¢ is dynamic if ¢ = dyn; otherwise ¢ is
static. The intention is that

e sta € Styp[nat] will tag expressions of type nat that are guaranteed to evaluate to known data (1.e., numerals
0, sc 0, ete.);

e dyn,,. € Styp[nat] will tag expressions of type nat whose evaluation may depend on unknown data and
thus cannot be guaranteed to evaluate to numerals;

® ¥, — or, € Styp[r — 7] will tag expressions of type 71 — 7 that are guaranteed to evaluate to known
data (1.e., abstractions);

o dyn__ € Styp[ri = 7] will tag expressions of type 71 — 73 whose evaluation may depend on unknown
data and thus cannot be guaranteed to evaluate to an abstraction.

Specialization types are encoded in EIf via the type family styp:typ —> type. For the dyn and barrow con-
structors, the type indices are encoded as logic variables T, T1, and T2, which will be instantiated by unification.
This formalizes the above convention of allowing type indices to be inferred from the context.

4.1.3 Binding-time logic

Figure 9 presents a natural-deduction-style logic for deriving judgements of the form bta e : 7 [w : ¢;]. Parenthe-
ses in the hypotheses of the rules for binding constructs (i.e., bta_lam_s, bta_lam_d, bta_fiz_s, bta_fiz_d) indicate

11



the discharging of zero or more occurrences of assumptions. We write [' - bta e : 7 [w : @] when bta e : 7 [w : ;]
is derivable under undischarged assumptions T'.

Derivable judgements specify constraints that an actual binding-time analysis algorithm must satisfy. Intu-
itively, if T & bta e : 7 [w : @], then given initial binding-time assumptions I', a binding-time analysis algorithm
may map e € Terms[A] of type 7 € Types[A] to a directive annotated term w € Terms[Ay:] of specialization type
@, € Spec-types[t]. Specifying the analysis in this way allows one to reason about correctness of the analysis
independently of the actual algorithm — proving correctness with respect to the constraints implies that any
algorithm satisfying the constraints will be correct. Furthermore, Section 5.1 shows how the Elf proof search
mechanism can be used to extract algorithmic content directly from the logic specifying the constraints.

We only consider assumptions involving identifiers (e.g., bta = : 7 [y : ¢,]) since the logic only allows dis-
charging of assumptions of this form. For T' = {bta z1: 7 [y1:¢nl, ..., bta n o [Yn : ©r, ]}, the z; are
required to be pairwise distinct (similarly for the y;). A static assumption (resp. dynamic assumption) is an
assumption bta @ : 7 [y : ¢-] where @, is static (resp. dynamic). I'; denotes a set of only static assumptions; I'q
denotes a set of only dynamic assumptions.

A simple induction over the structure of deductions for I' F bta e : 7 [w : ,] shows that the relation between
e and w is one-to-many, i.e., there may be many valid annotations of e. As a consequence, a binding-time
analysis satisfying these constraints has some flexibility in assigning correct annotations. In practice, one
desires an analysis that gives as many eliminable directives (i.e., gives as many s indices) as possible.* The
constraint system is complete in the sense that one may always obtain a valid annotation of a type correct e by
annotating all components as residual.

Finally, since the relation induced by bta is one-to-many from A to Ag,, it is many-to-one (i.e., a function)
from Ay to A. This is essentially an annotation forgetting function that removes directives and lift constructs.

The binding-time judgement is encoded in Elf as follows.

bta : exp->{t :typ}sexp->stypt -> type.

The dependent function type (z.e., {t : typ}...) expresses the dependency of the indexed specialization type
on the type of the A expression.

Figure 10 presents the Elf encoding of the binding-time logic. The implicit universal quantification of the
upper case variables captures the schematic nature of the rules. The rules for binding constructs (which involve
hypothetical proofs in the premises) are encoded in Elf as proof constructors that take proofs of hypothetical
Judgements as arguments. Such proofs are represented as functions mapping proofs of assumption judgements
to proofs of consequent judgements. The higher-order syntax representations requires that judgements involving
identifiers be represented as schematic judgements (i.e., identifiers are TI-quantified). This is the case in rules
bta_lam_s, bta_lam d, bta fix_s, and bta_fix d where e.g., the judgement bta (E x) T2 (W y) P2 expresses
that E and W may be instantiated to representations of terms with free occurrences of z and y respectively [19,
Section 3.1].

4.2 Specialization

4.2.1 Natural deduction rules

Figures 11 and 12 present the specialization logic for Ay terms.

e Figure 11 describes the evaluation of eliminable constructs. These rules correspond to the usual call-by-
name “natural” or “deductive” operational semantics for A given in Figure 4.

e Figure 12 describes the reconstruction of residual constructs after subexpressions have been specialized.
The rules for lamg and fixy are noteworthy because they introduce operations on open terms (e.g., bodies
of lam, and fixy constructs). Following Hannan [15, p. 144], we specialize the body of a lamy (and fix,)
under the assumption that the bound variable evaluates to itself.

4 We omit considering whether a term has a “best” annotation according to the logic. See the work on A-mix [10, 11] for related
discussions.

bta_z_s : bta 0 : nat [0, : sta]

bta_sc_s : bta e : nat [w : sta]

bta sc e : nat [sc; w : sta]

bta_pr_s : bta e : nat [w : sta]
bta pr e : nat [pr, w : sta]

bta_if_s : bta €1 : nat [wy : sta] bta e : T [wy : @] bta ez : 7 [wa : @]
bta if0 ey eq ez : 7 [if05 wi we w3 : @]

(btaz: 7 [y: 1))

btatam.s: bta e : 5 [w : o]
btalamaz.e:m — m [lamsy. w1 — o]
bta_app_s : bta eg : 1 — T2 [wo 1 1 — Pa] bta e1 : 7 [wy : 1]
bta app ege1 : T2 [app, wo w1 : 3]
. (btaz:7[y:¢])
blafirs btae: T [w: g
bta fixz.e : 7 [fixsy. w : @]
bta_z_d : bta 0 : nat [04 : dyn]
bta_sc_d : bta e : nat [w : dyn]
bta sc e : nat [scq w : dyn]
bta_pr_d: bta e : nat [w : dyn]

bta pr e : nat [pry w : dyn]

bta_if d:  bta e1 : nat [wy : dyn] bta e : T [ws : dyn] bta ez : T [ws : dyn]

bta if0 ey e e3 : 7 [if0; w1 we w3 : dyn]

(bta z : 7 [y : dyn])

btalam.d: bta e : T [w : dyn]
btalamz .e: 17 — 79 [lamgy . w : dyn]
bta_app_d : bta eq : 1 — T2 [wq : dyn] bta eq : 7 [wy : dyn]
bta app egey : T2 [app, wo w1 : dyn]
) (bta z : T [y : dyn])
btafiz.d: bta e : T [w : dyn]
bta fixz.e : 7 [fixqy. w : dyn]
bta_lift : bta e : nat [w : sta]

bta e : nat [lift w : dyn]

Figure 9: The binding-time logic

bta_z_s :

bta_sc_s :

bta_pr_s :

bta_if_s :

bta_lam_s

bta_app_s

bta_fix_s

bta_z_d :

bta_sc_d :

bta_pr_d :

bta_if_d

bta_lam_d :

bta_app_d :

bta_fix_d :

bta_lift :

bta z nat (bz s) sta.

bta (sc E) nat (bsc s W) sta
<- bta E nat W sta.

bta (pr E) nat (bpr s W) sta
<- bta E nat W sta.

bta (if E1 E2 E3) T (bif s W1 W2 W3) P
<- bta E1 nat W1 sta

<- bta E2 T W2 P

<- bta E3 T W3 P.

bta (lam E) (arrow T1 T2) (blam s W) (barrow P1 P2)
<- {x} {y} bta x T1 y P1 -> bta (E x) T2 (W y) P2.

bta (app EO E1) T2 (bapp s WO W1) P2
<- bta EO0 (arrow T1 T2) WO (barrow P1 P2)
<- bta E1 T1 W1 P1.

bta (fix E) (arrow T1 T2) (bfix s W) (barrow P1 P2)
<- {x} {y} (bta x (arrow T1 T2) y (barrow P1 P2))
-> (bta (E x) (arrow T1 T2) (W y) (barrow P1 P2)).

bta z nat (bz d) dyn.

bta (sc E) nat (bsc d W) dyn
<- bta E nat W dyn.

bta (pr E) nat (bpr d W) dyn
<- bta E nat W dyn.

bta (if E1 E2 E3) T (bif 4 W1 W2 W3) dyn
<- bta E1 nat W1 dyn

<- bta E2 T W2 dyn

<- bta E3 T W3 dyn.

bta (lam E) (arrow T1 T2) (blam d W) dyn
<- {x} {y} (bta x T1 y dyn) -> (bta (E x) T2 (W y) dyn).

bta (app EO E1) T2 (bapp d WO W1) dyn
<- bta E0 (arrow T1 T2) WO dyn
<- bta E1 T1 W1 dyn.

bta (fix E) (arrow T1 T2) (bfix d W) dyn
<- {x} {y} (bta x (arrow T1 T2) y dyn)
-> (bta (E x) (arrow T1 T2) (W y) dyn).

bta E nat (lift W) dyn
<- bta E nat W sta.

Figure 10: The EIf encoding of the binding-time logic

spec_z_s : 0s Uspee O

spec_sc_s : w spec a
scs W spec sCs @

spec_prz_s : w Yspec Os
pry w 'U'spec Os

spec_prsc_s w spec SC; @
pro w Yspec @

spec_ifz_s wi Jspec Os wy spec @
if0; w1 wa wz Yspec @

spec_ifsc_s w1 Yspec scs a1 w3 Yspec @
if0, w1 wa w3 Yspee @

spec_lam_s : lam, y.w Yspec lamgy.w

spec_app_s . wq Yspec lam, y . wy woly = wi] Yspee @

app, wo w1 Yspec @

spec_fir_s : wly = fix, y. w] Yspec a
fix, y.w ‘U'spec a

Figure 11: The specialization logic (static constructs)

spec_z_d : 04 Uspec 04

spec_sc_d : W spec @
scq W {Jspec SCq @

spec_pr_d : W spec @
prqg w 'U'spec prqga

5]76’(3—77](—611 w1 'U'spec aj wa ‘U'spec az w3 ‘U'spec as
if0q w1 wa w3 Yspee if04 a1 as as

(y Vspec ¥)

spec_lam_d : w 1) a
spec

lamgy . w {spec lamgy. a

spec_app_d : wo Jspec @0 w1 Yspec a1

appg wo Wy ‘U’spﬁc appg @o a1
spec_fizx_d : (v spec ¥)
w ‘U'spec a

fixgy. w Yspec fixay.a
spec_lift : w Yspec 05
lift w Yspec 04
spec_lift_sc w Yspec scga’ ifta’ Ygpee a

lift w {spec scqa

Figure 12: The specialization logic (dynamic and coercion constructs)

spec_z_s : spec (bz s) (bz s).

spec_sc_s : spec (bsc s W) (bsc s A)
<- spec W A.

spec_prz_s : spec (bpr s W) (bz s)
<- spec W (bz s).

spec_prsc_s : spec (bpr s W) A
<- spec W (bsc s A).

spec_ifz_s : spec (bif s W1 W2 W3) A
<- spec W1 (bz s)
<- spec W2 A.

spec_ifsc_s : spec (bif s W1 W2 W3) A
<- spec W1 (bsc s A1)
<- spec W3 A.
spec_lam_s : spec (blam s W) (blam s W).
spec_app_s : spec (bapp s WO W1) A
<- spec WO (blam s WO’)
<- spec (WO’ W1) A.
spec_fix_s : spec (bfix s W) A
<- spec (W (bfix s W)) A.

Figure 13: The Elf encoding of the specialization logic (static constructs)

spec_z_d : spec (bz d) (bz d).

spec_sc_d : spec (bsc d W) (bsc d 4)
<- spec W A.

spec_pr_d : spec (bpr d W) (bpr 4 A)
<- spec W A.

spec_if_d : spec (bif d W1 W2 W3) (bif d A1 A2 A3)
<- spec W1 A1l
<- spec W2 A2
<- spec W3 A3.

spec_lam_d : spec (blam d W) (blam 4 A)
<- {x} spec x x -> spec (W x) (A x).

spec_app_d : spec (bapp d WO W1) (bapp d AO A1)
<- spec WO AO
<- spec W1 A1l.

spec_fix_d : spec (bfix d W) (bfix 4 A)
<- {x} spec x x -> spec (W x) (A x).

spec_lift : spec (lift W) (bz d)
<- spec W (bz s).

spec_liftsc : spec (lift W) (bsc d A)

<- spec W (bsc s A?)
<- spec (1lift A’) A.

Figure 14: The Elf encoding of the specialization logic (dynamic and coercion constructs)

An alternative to introducing local assumptions for bound variables would be to add the axiom y {spec
y (which states that all variables evaluate to themselves). However, this approach cannot be directly
represented in EIf. Under the higher-order abstract syntax representation, the variables y are not objects
in the Elf signature. The approach taken in Figure 12 seems the most natural with the higher-order
abstract syntax representation.

e The final rules of Figure 12 coerce an eliminable numeral to a residual expression.

We write ¥ F w Yspec @ when w spec a is derivable under undischarged assumptions 3. Intuitively, if
Y F w Yspec @, then the specializer maps w € Terms[Ay] to answer a € Terms[Ap] in context 3. We only
consider assumptions involving identifiers (e.g., y {spec y) since the logic only allows discharging of assumptions
of this form. For ¥ = {y1 Uspec ¥1, -, Yn Uspec ¥n} the y; are required to be pairwise distinct. X is compatible

withT = {btazy 7 [y1:¢n], ., btaz, 7 [yn @7 ]} IS = {y1 Yspee Y1, -y Un Yspec Yn}. Tt is easy to
check that the relation induced by {};pe. is a partial function (given the constraints on assumptions above).

4.2.2 Elf encoding

Figures 13 and 14 give the Elf encoding of the specialization logic. The encoding techniques are similar to those
used in the previous section. The specialization judgement specifies a binary relation (which is actually a partial
function) on annotated terms Sezp and is encoded in EIf as follows.

spec : sexp —> sexp —> type.

In the rules spec_app_s and spec_fiz_s, we take advantage of the higher-order abstract syntax representation
and use Elf application (z.e., f-reduction) to implement capture-free substitution.

4.2.3 Canonical terms of specialization

In the conventional deductive semantics for A in Section 3.3, one has numerals (e.g., 0, sc 0) and lamz .¢ as
canonical terms, z.e., these terms are the results of evaluation. This was formalized by the val judgement.

Intuitively, a specializer is part evaluator and part compiler. Therefore, the canonical terms of the special-
izer are numerals and lam, y.w (corresponding to evaluation results), and completely residual terms or code
(corresponding to compilation results). In a manner analogous to the formalization of values (canonical terms
of evaluation), Figure 15 formalizes the notion of answers (canonical terms of specialization) by defining a
judgement ans w : @;.

We write A F ans w : ¢, when ans w : ¢, is derivable under undischarged assumptions A. A is compatible
withT = {btazy:7 [y1:¢0n], 0 bta zy T [yn s, ]} if A = {ans y1 : @r,, ..., ans y, : ¢, }. Intuitively,
if ans w : ¢; holds, then w is a canonical term of specialization type ¢,. In particular, if + ans w : dyn, then
w is completely residual.

In contrast with the wal judgement of Section 3.3 which contained no notion of typing, the judgement
ans w : @, relates a canonical term w to a particular specialization type ¢,. We will need this additional
information when stating the correctness of binding-time analysis in Section 6.1. As with the wal judgement,
the ans judgement admits answers that are not well-annotated. For example, F anssc(lamz.z):dyn, but
there does not exist e and 7 such that F bta e : 7 [sc (lama . ) : 7]. Well-annotatedness of answers is expressed
by using the binding-time judgement in conjunction with the answer judgement. Figure 16 gives the Elf encoding
of the answer logic.

In Section 3.3, Proposition 1 showed that the value judgement val described the results of evaluation. An
analogous proposition below shows that the answer judgement describes the results of specialization.

Proposition 2 If - w {,pec @ and = btae:7[w:p,], then - ansa: ¢, .

The proof of this proposition follows as a corrollary of the mechanically verified proof of Theorem 1 in
Section 6.

ans_z_s .

ans_sc_s :

ans_lam_s :

ans_z_d :

ans_sc_d :

ans_pr_d :

ans_if-d :

ans_lam_d :

ans_app_d :

ans_fix_d :

ans 0, : sta

ans w : sta
ans scg w : sta

ans lamgs z . w : o1 — P3
ans 0g : dyn

ans w : dyn
ans scg w : dyn

ans w : dyn
ans pryw : dyn

ans wy : dyn ans ws : dyn ans ws :

ans if0q wy wo w3 : dyn

(ans x : dyn)
ans w : dyn
ans lamgz . w : dyn

ans wq : dyn ans wy : dyn

ans appg wo wy : dyn

(ans @ : dyn)
ans w : dyn
ans fixqg z.w : dyn

Figure 15: The answer logic

%%% answer judgement

ans : sexp -> styp T -> type.
%%% answer logic

% static answers

ans_z_s : ans (bz s) sta.

ans_sc_s : ans (bsc s W) sta
<- ans W sta.

ans_lam_s : ans (blam s W) (barrow P1 P2).
% dynamic answers
ans_z_d : ans (bz d) dyn.

ans_sc_d : ans (bsc d W) dyn
<- ans W dyn.

ans_pr_d : ans (bpr d W) dyn
<- ans W dyn.

ans_if_d : ans (bif 4 W1 W2 W3) dyn
<- ans W1 dyn
<- ans W2 dyn
<- ans W3 dyn.

ans_lam_d : ans (blam d W) dyn
<- {x} ans x dyn -> ans (W x) dyn.

ans_app_d : ans (bapp d WO Wi1) dyn
<- ans WO dyn
<- ans W1 dyn.

ans_fix_d : ans (bfix d W) dyn
<- {x} ans x dyn -> ans (W x) dyn.

Figure 16: The Elf encoding of the answer logic

4.3 Partial evaluation

We outline how the logics above define offline partial evaluation using the following object term.

def
e = app (lamz1 . app x5 (app (lam x5 . 23) 1)) xo

The free variables zq, x5 represent input parameters. The assumptions 'y = {bta ¢ : nat [y, : sta]} and Ty =
{bta x5 : nat — nat [y, : dyn]} identify zo of type nat as known and z5 of type nat — nat as unknown. They also
associate zg and x5 with annotated language identifiers yo and y5. The fact that T'y U T'y F bta e : nat [w : dyn]
where

def .
w = app, (lam, y1 . app, y2 (lift (app, (lam; ys . ys) y1))) ¥o

expresses that a binding-time analysis may associate e with w based on assumptions T'y; U T'4.
To prepare for specialization, we supply known input via substitution.

T4t bta e[zg := 0] : nat [wlyo := 05] : dyn]

Now taking 3 = {y2 {lspec y2} compatible with T'y (expressing that the dynamic parameter evaluates to itself),
the specialization logic gives

Y wlys = 0s) Yspec appqy2 04.
Theorem 1 (correctness of binding-time analyis) of Section 6.1 tells us there exists ¢ € A such that
'y F bta €' :nat[appy;y204 : dyn]. As noted in Section 4.1, ¢/ must be appx20 — the unannotated version

of app, y2 04. Theorem 3 (soundness of specialization) of Section 6.2 tells us that e[z := 0] is convertible to ¢’
(denoted e[z := 0] =, €) in the program calculus for A. Thus, for all closed inputs d € A of type nat — nat,

elrg : =0, 29 :=d] =5 €'[zy := d].

This reflects the correctness criteria for partial evaluation given at the beginning of Section 4: running e on
static input 0 and dynamic input d is operationally equivalent to running ¢’ on d.

5 Prototyping an Offline Partial Evaluator for A

5.1 Prototyping a binding-time analysis

The Elf encoding of the binding-time logic (see Figure 10) gives a prototype of the binding-time analysis. The
following EIf query returns all the possible annotations that may be assigned to the example term of Section 3
(here we make dyn the specialization type of the entire term).®

?- bta (app (lam [x1] lam [x2] app (lam [x3] x3) x1) z) (arrow nat nat) W dyn.
Solving...

W = bapp s
(blam s ([yl:sexp] blam d ([y2:sexp] bapp s (blam s ([y3:sexp] y3)) (lift y1))))
(bz s).

=
[}

bapp s
(blam s ([yl:sexp] blam d ([y2:sexp] lift (bapp s (blam s ([y3:sexp] y3)) y1))))
(bz s).

There are actually twelve correct annotations; we show only the two above.

5.2 Prototyping a specializer

The EIf encoding of the specialization logic (see Figures 13 and 14) gives a prototype of the specializer. The
following EIf query returns the result A of specializing the second annotated term above (there is only one answer
since spec is a function).

5Some of the results of EIf evaluation are a-converted for clarity.

?- S : spec (bapp s
(blam s ([yl:sexp] blam d ([y2:sexp] lift (bapp s (blam s ([y3:sexpl y3)) y1))))
(bz s)) A.

Solving...

A
S

blam d ([y2:sexp] bz d),

spec_app_s
(spec_lam_d [y2:sexp] [S:spec y2 y2] spec_lift (spec_app_s spec_z_s spec_lam_s))

spec_lam_s.
H

no more solutions

In this query, we add the variable S which binds to the specialization deduction used to obtain 4.°

6 Verifying an Offline Partial Evaluator for A

6.1 Binding-time analysis

A binding-time analysis is correct if it always produces consistent specialization directives. Directives are
consistent if the specializer can never “go wrong”. As described by Palsberg [27], a specializer may go wrong
for two reasons: 1) it trusts a part of the program to be eliminable when in fact it is residual, and 2) it trusts
a part of the program to be residual when in fact it is eliminable.

The specializer goes wrong for the first reason on app, (lamg 2 . 2) 05. In this case, it attempts evaluation using
the rule spec_app_s (see Figure 11) but “hangs” (i.e., the result of specialization is undefined) since lamgz . z is
not an eliminable abstraction. Similar problems may occur with the rule spec_lift (e.g., if w {spec 0a).

The specializer goes wrong for the second reason on app, (lam, z.z)04. In this case, it attempts residual-
ization using the rule spec_app_d and incorrectly produces an output program that is not completely residual
(since lamg z . z is eliminable).

The property that the specializer never goes wrong (i.e., the binding-time analysis always yields consistent
directives) is a generalization of the type-soundness property for standard type systems. To establish type-
soundness, one typically proves a subject-reduction result showing that typing is maintained under evaluation.
To establish consistency of directives, we prove that specialization typing is maintained under specialization.
Furthermore, we show that specialization results are always answers of the appropriate specialization type, z.e.,
that the ans is always satisfied (as promised in Section 4.2).

This ensures that the specializer will not go wrong for the first reason. For example, in the rule spec_app_s,
the rule bta_app_s guarantees that wq always has specialization type @1 — @2 and if wo Yspec ao then aqg =
lam; z . w, since only appropriate answers are produced.

This also ensures that the specializer will not go wrong for the second reason. For example, in the rule
spec_app-d, the rule bta_app_d guarantees that wo always has specialization type dyn and if wg Uspec @0 then ag
must be completely residual (similarly for a1).

Theorem 1 (Correctness of binding-time analysis) If X F w Yspec a and Tg b bta e : 7 [w: @,] and T is
compatible with T4, then there exists e’ € A such that Tq & btae' :7[a:¢,] and A F ans a: o, where A is
compatible with T 4.

Proof: For notational convenience, we write D :: J when D is a deduction of judgement 7 and state deduction
rules in a linear format. For example, a specialization deduction that ends with the rule spec_app_d is written
spec_app_d(So, S1) 1 appy wo w1 Yspec appg ag ag where it w; Yypee a; (i = 1,2). The notation is somewhat
imprecise since we do not use an explicit discharge function for assumptions [19, Section 4.1]. However, the EIf
encoding makes matters sufficiently clear.

For the theorem hypotheses, let S 1 w spec @ and B :: btae: 7 [w: ¢,]. We show an effective method
for constructing deductions C :: bta e’ : 7 [a: ¢;] and D :: ans a : ¢, where the compatibility constraints on
undischarged assumptions are satisfied. The proof proceeds by induction on the pair of deductions § and B,

6Note that the order of arguments to the deduction constructors is the reverse of what one might expect since we use <- (instead
of =>) in the encodings of the binding-time and specialization logics.

i.e., the method is primitive recursive. Although this primitive recursive method cannot be represented in EIf
as a function (since it is not schematic), it can be represented as a relation via the following judgement.

tl : spec WA ->btaET WP ->btaE’TAP ->ans AP -> type.

Each case of the constructive proof is formalized as a rule for the t1 judgement. Below we show three illustrative
cases (each increasing in complexity).
case § = spec_z_s 1 05 spec 0; and B = bta_z_s :: bta 0 : nat [0, : sta]:

The required deductions are bta_z_s :: bta 0 : nat [0, : sta] and ans_z_s :: ans Q, : sta. This is formalized by
the following axiom.

ti_z_s : tl (spec_z_s) (bta_z_s) (bta_z_s) (ans_z_s).

case S = spec_app_d(So, S1) 1 appy wo w1 Yspec appy do a1
B = bta_app_d(Bq , B1) :: bta appeg ey : 79 [appy wo wy : dyn] :

Applying the inductive hypothesis to Sp and By gives deductions Cq :: bta e : 1 — 72 [ag : dyn] and Dq ::
ans ag : dyn. Applying the inductive hypothesis to S and By gives deductions Cy :: bta €} : 71 — 7 [a1 : dyn]
and Dy :: ans a; :dyn. The required deductions are bta_app_d(Co, C1) and ans_app_d(Do, D1). This is
formalized by the following rule (the arguments to the proof constructors appear in reverse order (e.g.,
spec_app_d S1 SO instead of spec_app_d SO S1) due to the use of <~ (instead of =>) in the encodings of
binding-time and specialization logic).

t1l_app_d : t1 (spec_app_d S1 S0) (bta_app_d B1 BO)
(bta_app_d C1 C0O) (ans_app_d D1 DO)

<- t1 SO BO CO DO

<- t1 S1 B1 C1 D1.

In operational terms, the inductive hypotheses are manifested as recursive calls to the function represented
by t1.

case S = spec_app-s(So, S1) 1 app, wo w1 Yspec @
B = bta_app_s(Bo, B1) :: bta appeger : ™ [app, wo w1 : 2]

Applying the inductive hypothesis to Sg and By gives deductions Cf :: bta ef : 71 — 72 [ao : ¢1 — 2] and

Dy = ansag: @1 — p2. An examination of the ans rules (Figure 15) reveals that ag = lam,y. wy.
Then, an examination of the axiomatization of the bta judgement (Figure 9) shows that we must have
Cy = btalam_s(Cq :: bta eff = 7 [wh @ @2]) :: bta lamx . e” 1 7 — m [lamy y. wy : p1 — 2] where the undis-

charged assumptions of Cp include bta x : 7 [y : ¢1]. A simple substitution lemma (which we omit) allows
us to replace each of these assumptions in Cy with the deduction By :: bta eq : 7y [wy : 1] to obtain a deduc-
tion By :: bta ef[z 1= e1] : 7 [wily := wi] : ¢a]. Applying the inductive hypothesis to S; and By gives the
required deductions C :: bta €’ : 75 [a : p3] and D :: ans a : 5. This is formalized as follows.

tl_app_s : t1 (spec_app_s S1 SO) (bta_app_s B1 BO) C D
<- t1 SO BO (bta_lam_s C0O) DO
<- t1 81 (CO _ _ B1) CD.

The “examination of the rules” (that told us deduction C, must have bta_lam_s as its last rule) is captured by
matching (i.e., (bta_lam_s €0)). The required substitution lemma appears for free due to the higher-order
abstract syntax representation and the Transitivity derived rule of the LF calculus [19, Section 2.3]; it is
captured by (€0 _ _ B1). The underscores correspond to the terms e; and w; above. Using the underscores
and letting EIf reconstruct allows us to encode the rule more concisely.

The complete set of rules is given in Appendix A. |

The following EIf query illustrates the function induced by the t1 rules. Given the specialization deduction
S of Section 5.2 and a binding-time deduction for the term being specialized (i.e., the second annotated term
of Section 5.1), t1 constructs a binding-time deduction C for the specialization answer as well as an answer
deduction D proving that the answer is completely residual.

?- t1 (spec_app_s
(spec_lam_d [y2:sexp] [S:spec y2 y2] spec_lift (spec_app_s spec_z_s spec_lam_s))
spec_lam_s)
(bta_app_s bta_z_s
(bta_lam_s [x1:exp] [yl:sexp] [Bl:bta x1 nat yl sta]
bta_lam_d [x2:exp] [y2:sexp] [B2:bta x2 nat y2 dyn]
bta_lift
(bta_app_s B1
(bta_lam_s [x3:exp] [y3:sexp] [B3:bta x3 nat y3 sta] B3))))

C D.
Solving...
D = ans_lam_d [y2:sexp] [D2:ans y2 dyn] ans_z_d,
C = bta_lam_d [x2:exp] [y2:sexp] [B2:bta x2 nat y2 dyn] bta_z_d.

Elf type-checking mechanically verifies that deductions € and D are well-formed when they exist. However,
verification that such deductions always exist (i.e., that t1 is total) cannot be captured in EIf. This phase of
verification (called schema checking) must be done by hand, although its automation is the subject of current
research [32]. Our definition makes t1 total because we give a rule for each possible pair of spec and bta rules
(giving primitive recursive structure). In addition, the deduction of answer judgements tell us that matching is
used (in rules t1_app_s and t1_1ift) only when it will always succeed.

6.2 Soundness of specialization

A specializer is sound if its steps describe a meaning preserving transformation on the unannotated object
program. Specifically, if (a) e; is the unannotated version of the program to be specialized (with the static data
already supplied), and (b) es is the unannotated version of the result of specialization, then a sound specializer
should guarantee that e; and ey are operationally equivalent (as defined in Section 3.3).

This ensures that e; and ey will behave the same under evaluation when dynamic data is supplied. Specif-
ically, for all dynamic data d such that T' F ¢pei[x := d] :nat and T & tpes[z := d] : nat, e1[z == d] ~
ea[z := d]. The last claim follows since the substitution [z := d] can be represented with the context
app (lamz . [']) d and the fact that & is easily shown to be a congruence.

We prove the operational equivalence of e1 and es by appealing to a program calculus for A (i.e., an equational
theory for deducing operational equivalences). Gunter [14, Chapter 4] gives a calculus for PCF, and Figure 17
recasts the calculus in our setting. e; =, es is the judgement expressing convertibility in the calculus. The
rules are generally the expected ones for call-by-name. However, note the use of the judgement val in the
rules conv_prsc and conv_ifsc. This reflects the strictness of the primitive operation pr and the strictness of the
conditional ifQ in the test position. Appendix B.1 gives the EIf encoding of the calculus. Tt is similar to the
encoding given by Pfenning for the pure untyped A-calculus [30].

The following theorem expresses the soundness of the calculus for reasoning about A operational equivalences.

Theorem 2 (Soundness of calculus) For all e1,e5 € A such that TFiper 7 and TF tpes: T,
€1 = es 1mplies e1 & es.

Existing techniques for proving the theorem are fairly involved. For example, Gunter [14] appeals to an
adequacy result connecting the operational semantics and a model. Other techniques use co-induction principles
[12]. Unfortunately, it seems impossible to encode any of these techniques in EIf (Section 8 discusses the encoding
of these techniques in other frameworks).

The inability to encode the proof in Elf does not seem detrimental to our approach. Historically, one of
the reasons for introducing for a program calculus is to avoid repeated appeals to direct proofs of operational
equivalences and reason instead using the simple rules of the calculus. The appropriateness of this approach is
underscored by the situation we face here — it is much easier to encode a proof that appeals to the calculus.

The following theorem captures the fact that the unannotated object program e is operationally equivalent
to the unannotated specialization result e’.

Figure 17: Program calculus for A

Theorem 3 (Soundness of specializer) If & F w Uspec @ and Tq F btae: 7[w: ¢;] and ¥ is compatible
with T4, then there exists ¢! € A such that Ty b btae' :7[a: p,;] ande =, €.

Proof: To formalize the proof, we construct a function (via a relation as in Theorem 1) which builds a deduction
showing e converts to €'

t3 : spec WA ->btaETWP ->btaE’TAP ->ans AP -> conv E E’ -> type.

The t3 judgement expresses a function which takes two proofs as input: a proof of spec W A and a proof of
bta E T W P. The output judgement bta E* T A P acts as the unannotating function from Ap; terms A to A
terms E’ (see Section 4.1.3).

The output proof of ans A P is necessary for extracting value deductions to be used in the conv_prsc and
conv_ifsc rules of the calculus. This requires a small lemma — essentially stating that if the specialization
type P is static, then dropping the annotations on the answer A yields a value. Tn other words, given a proof
bta E T W P, a proof ans W P that Wis a value, and a proof that P is static, we can construct a proof that the
de-annotation of W (given by E) is a value. This is formalized with the following judgement.

ans_to_val : bta ET WP -> ans W P -> static P -> val E -> type.

This proof is simple and omitted. The EIf encoding is given in Appendix B.2 along with the encoding of
predicates describing static and dynamic specialization types.

Finally, the conv E E’ judgement expresses the desired property that E converts to E’. A strategy similar
to the one used in the proof of Theorem 1 gives the rules defining t3 (see Appendix B.3).

Note that by including bta E> T A P and ans A P as we generate the deduction of conv E E’, we are in
essence reproving Theorem 1. An alternative approach would be to use Theorem 1 to justify some of the steps
in the proof below. This would be manifested in the Elf encoding as “calls” to the goal represented by the t1
judgement. We choose the formalization below to simplify the presentation. |

7 Related work

Despeyroux first emphasized using deductive systems to define transformations [8]. She specified a compiler, and
source and target language semantics using deductive systems. The specifications were executed via encodings
into Typol. Informal proofs of correctness were given as relations between deductions. However, these proofs
could not be formalized in Typol because it does not support the direct manipulation of its own deductions.

Hannan and Pfenning [18] improved upon this by formalizing similar proofs of correctness in Elf. EIf (unlike
e.g., Typol, and A-Prolog) supports the direct manipulation of its own deductions. Furthermore, EIf type
checking mechanically verifies that deductions corresponding to correctness proofs are well-formed. Other work
using Elf to encode operational semantics includes Michaylov and Pfenning’s encoding of call-by-value natural
semantics and associated meta-theory for the language Mini-ML [23], Pfenning’s representation of a Church-
Rosser proof [30], Hannan’s encoding of a type system for closure conversion [16], and Niss and the author’s
representation of the correctness proofs for the familiar translation of call-by-name to call-by-value using thunks
[26].

Despeyroux [8, Section 8], and Hannan and Pfenning [18, Section 7], suggested that their methods could be
used to specify “mixed computation” and partial evaluation, respectively. Hannan and Miller [17] carried out
Despeyroux’s suggestion; they use deductive systems encoded in A-Prolog to obtain executable specifications of
mixed computation (in their work, “mixed computation” = non-deterministic on-line partial evaluation).

Our contributions include using deductive systems to specify program specialization directed by type-based
analysis. Moreover, we adapt the techniques of Hannan and Pfenning [18], and Michaylov and Pfenning [23] to
mechanically verify correctness.

In doing so, we obtain verified proofs similar in scope to the ones given by Gomard and Jones [10, 11] for
A-mix. Their denotational meta-language is significantly more complex than our logic-based meta-language and
it 1s unlikely that the proofs there could be formalized or mechanically verified to the extent that we have done
here. However, it must be noted that one of their goals was self-application. Relying on the similarity between
their meta-language and object-language, they obtain an object-language specification of the partial evaluator

(giving self-applicability) by a fairly easy (though informal and unverified) translation from the meta-language
specification. Tn our setting this is more difficult since the character of our meta-language (logical) is quite
different from our object language (functional).

Building upon the work of Gomard, Jones, and Mogensen, Palsberg [27] and Wand [37] give detailed pre-
sentations of the correctness of binding-time analysis. Palsberg presents a generalization of Gomard and Jones
criteria [11] for consistent binding-time annotations. Wand studies Mogensen’s self-applicable partial evaluator
[24] for the pure A-calculus. His binding-time analysis is essentially the same as Gomard and Jones’s, as well
as the one presented here. Wand’s goals with respect to correctness of the analysis and specializer are more
ambitious than those here, because his correctness criteria is strong enough to specify the behaviour of the
partial evaluator when self-applied.

It would be interesting to see to what extent the meta-theory used by Palsberg and Wand could be formalized
in Elf. In particular, once Wand has established the correctness results for an interpreter and partial evaluator,
his proofs for the Futamura projections have a simple calculational style (based on equational reasoning). The
proofs follow the same steps as the “Mix equations” [20, Chapter 1] which are often used to reason about the
results of self-application. We are currently investigating how these reasoning principles can be formalized (e.g.,
to obtain a “Mix equational theory”) and encoded in LF.

In recent work, Davies and Pfenning [7] give a type system for expressing staged computation based on
the intuitionistic modal logic S4. They have implemented the type system and a portion of the associated
correctness proofs in Elf.

Our work focuses on partial evaluation of functional programs, but similar techniques can be applied to
imperative languages as well. Blazy and Facon [2] specify a simple specializer for FORTRAN using natural
semantics, and derive a prototype from the specification using the Centaur programming environment. However,
their correctness proofs are not formalized since the Centaur environment is not oriented toward this purpose.
Bertot and Fraer [1] show how similar correctness proofs (for a specializer for an imperative language) can be
formalized and mechanically checked using Coq.

8 Conclusion

8.1 Summary

We have specified the main components of a simple offline partial evaluator (i.e., binding-time constraints and
a program specializer) using natural-deduction style logics. These specifications were formalized by encoding
them in LF. Prototypes were obtained directly from the formal specifications using Elf. We formalized and
mechanically verified a significant portion of the meta-theory of offline partial evaluation (e.g., correctness of
binding-time analysis and soundness of the specializer) via meta-programming in EIf. A certain degree of
synergy is obtained by using the declarative formalism of deductive systems: one may specify, prototype and
mechanically verify correctness via meta-programming — all within a single framework.

8.2 Assessment

Using deductive systems to specify analyses and transformations: The simplicity of the specifications
stems from their declarative nature (which allows inessential algorithmic details and model theory to be avoided).
This does not mean that other formalisms such as denotational semantics are not useful. There are certainly
situations where the extra reasoning capabilities offered by model theory are beneficial. However, it seems that
in many applications — particularly in the area of program specialization — these extra capabilities are never
utilized and other styles of formalization such as the ones used here may be more appropriate.

Using LF as a meta-language for defining specifications: Using a formal meta-language such as LF
forces one to be precise, and such precision usually lends clarity to presentations. However, the rigidness of
LF almost demands that one cast specifications as natural-deduction-style logics and use higher-order abstract
syntax. This style of logic may not be appropriate in all situations. In addition, it would be interesting to see
how our specifications and proofs would be formalized in other logical frameworks. Recent work suggests that it
may be possible to generate program analyses from formal specifications. LF stands as a possible meta-language
for formalizing such specifications.

Using EIf to obtain executable specifications: We found EIf particularly useful for prototyping the
binding-time analysis. It should be investigated if this scales up to more complicated type-based analyses that
use conjunctive types and more general forms of subtyping. More complex specifications may well be prone
to looping due to the simplistic depth-first search strategy used by Elf. However, we have successfully used
the techniques presented here to prototype a more complex higher-order version of the multi-level binding-time
analysis of Gliick and Jgrgensen [9].

Using EIf to mechanically verify correctness: It seems possible to use these techniques to verify other
forms of program specialization. For example, if one takes the view that an annotated program is its own
generating extension, the techniques here are very close to what one would use to verify the correctness of a
hand-written cogen [20]. Tt remains to be seen if these techniques scale up to more robust forms of partial
evaluation that include sophisticated folding strategies. One approach might be to use Sand’s calculus for sound
fold /unfold transformations [34].

The logic of EIf was not strong enough to encode the calculus soundness proof (see Theorem 2). More
generally, the induction principles as required in the proofs of Theorems 1 and 3 could not be completely
formalized. This manifested itself in the inabilty to mechanically guarantee the totality of the functions (defined
as relations) representing the constructive proofs. Frameworks based on stronger logics (such as Isabelle-HOL
and Isabelle-ZF [28], HOL [13], and Coq) can overcome these problems. Nipkow [25] has used Isabelle-HOL to
encode a large portion of the theory given in Winskel’s semantics textbook [39]. Similar techniques would be
used in formalizing the calculus-soundness proof using model theory. Collins [4] has used HOL to encode a theory
of bi-similarity for higher-order functional language. This could easily be extended to prove the soundness of the
calculus using coinduction techniques. Finally, the forthcoming thesis of Welinder [38] contains a formalization
of various principles required for reasoning about specialization of first-order functional programs.

Using stronger logics leads to tradeoffs. Proofs become much more involved (as formalization is more
complete). Prototypes are harder to obtain directly from specifications.

Correctness of binding-time analysis
definition of t1 judgement
spec WA ->bta ETWP ->btaE”TAP ->ans A P -> type.

definition of t1 axiom and rules

% static constructs

ti_z_s : tl (spec_z_s) (bta_z_s) (bta_z_s) (ans_z_s).

ti_sc_s : tl (spec_sc_s S) (bta_sc_s B) (bta_sc_s C) (ans_sc_s D)

<- t1 S B CD.

tl_prz_s : tl (spec_prz_s S) (bta_pr_s B) (bta_z_s) (ans_z_s)

<- t1 S B (bta_z_s) D.

tl_prsc_s : t1 (spec_prsc_s S) (bta_pr_s B) CD

<- t1 S B (bta_sc_s C) (ans_sc_s D).

ti1_ifz_s : t1 (spec_ifz_s S2 S1) (bta_if_s B3 B2 B1) C2 D2

<- t1 81 B1 (bta_z_s) D1
<- t1 S2 B2 C2 D2.

ti_ifsc_s : tl1 (spec_ifsc_s 83 S1) (bta_if_s B3 B2 B1) C3 D3

<- t1 S1 B1 (bta_sc_s C1) D1
<- t1 83 B3 C3 D3.

ti_lam_s : t1 (spec_lam_s) (bta_lam_s B) (bta_lam_s B) (ans_lam_s).

tl_app_s : tl (spec_app_s S1 S0) (bta_app_s B1 BO) C D

<- t1 SO0 BO (bta_lam_s CO) DO
<- t1 81 (CO _ B1) C D.

ti_fix_s : tl1 (spec_fix_s S) (bta_fix_s B) C D

<- t1 S (B (fix E) (bfix s W) (bta_fix_s B)) C D.

% dynamic constructs

ti_z_d : t1 (spec_z_d) (bta_z_d) (bta_z_d) (ans_z_d).

ti_sc_d : t1 (spec_sc_d S) (bta_sc_d B) (bta_sc_d C) (ans_sc_d D)
<- t1 S B CD.

ti_pr_d : tl1 (spec_pr_d S) (bta_pr_d B) (bta_pr_d C) (ans_pr_d D)
<- t1 S B CD.

ti1_if_d : t1 (spec_if_d S3 S2 S1) (bta_if_d B3 B2 B1)
(bta_if _d4 C3 C2 C1) (ans_if_d D3 D2 D1)
<- t1 S1 B1 C1 D1
<- t1 S2 B2 C2 D2
<- t1 S3 B3 C3 D3.

ti_lam_d : t1 (spec_lam_d S) (bta_lam_d B) (bta_lam_d C) (ans_lam_d D)
<- {x} {y} {S’: spec y y} {B’ : bta x T1 y dyn} {D’: ans y dyn}
t1 S’ B> B’ D’
->t1 (Sy$S’) (BxyB’) (CxyB’) (DyD’).
til_app_d : tl1 (spec_app_d S1 S0) (bta_app_d B1 BO)
(bta_app_d C1 CO) (ans_app_d D1 DO)
<- t1 SO BO CO DO
<- t1 S1 B1 C1 D1i.
t1_fix_d : t1 (spec_fix_d S) (bta_fix_d B) (bta_fix_d C) (ans_fix_d D)
<- {x} {y} {S’: spec y y}
{B’ : bta x (arrow T1 T2) y dyn} {D’: ans y dyn}
t1 S’ B’ B’ D’
->t1 (SyS’) (BxyB’) (CxyB’) (DyD’).

t1_1ift : t1 (spec_lift S) (bta_lift B) (bta_z_d) (ans_z_d)
<- t1 S B (bta_z_s) D.

t1_liftsc : t1 (spec_liftsc S2 S1) (bta_lift B1) (bta_sc_d C2) (ans_sc_d D2)

<- t1 S1 B1 (bta_sc_s C1) D1
<- t1 82 (bta_lift C1) C2 D2.

B Soundness of specializer

B.1 Convertability relation

%h% convertability judgement

conv : exp —> exp -> type.

%%k convertability axioms and rules

% axioms

conv_beta : conv (app (lam EO) E1) (EO E1).
conv_fix : conv (fix E) (E (fix E)).
conv_prz : conv (pr z) z.

conv_prsc : conv (pr (sc E)) E
<- val E.

conv_ifz :

conv_ifsc :

conv (if z E2 E3) E2.

conv (if (sc E1) E2 E3) E3
<- val El.

% inference rules

conv_refl

conv_sym :

conv_trans :

conv E E.

conv E1 E2
<- conv E2 E1.

conv E1 E3
<- conv E1 E2
<- conv E2 E3.

% context rules

conv_sc_c:

conv_pr_c:

conv_if_cl:

conv_if_c2:

conv_if_c3:

conv_lam_c:

conv_fix_c:

conv (sc E) (sc E?)
<- conv E E’.

conv (pr E) (pr E’)
<- conv E E’.

conv (if E1 E2 E3) (if E1’ E2 E3)
<- conv E1 E1’.

conv (if E1 E2 E3) (if E1 E2’ E3)
<- conv E2 E27.

conv (if E1 E2 E3) (if E1 E2 E3’)
<- conv E3 E3’.

conv (lam E1) (lam E2)
<- {x} (conv (E1 x) (E2 x)).

conv (fix E1) (fix E2)
<- {x} (conv (E1 x) (E2 x)).

conv_app_c0: conv (app EO E1) (app EO’ E1)

<- conv EO EO’.

conv_app_cl: conv (app EO E1) (app EO E1’)

B.2 Extracting value deductions from answer deductions

<- conv E1 E1’.

B.2.1 Predicates classifying specialization types

%h% judgements

static : styp T -> type.

dynamic : styp T -> type.

%%%h static

static_nat :

rules

static sta.

static_arrow : static (barrow P1 Pw).

%%% dynamic rules

dynamic_dyn : dynamic dyn.

B.2.2 Function from static answers to values

%%% ans_to_val judgement

ans_to_val : bta ET WP -> ans W P -> static P -> val E -> type.
%%% ans_to_val rules

ans_to_val_z_s : ans_to_val bta_z_s ans_z_s static_nat val_z.

ans_to_val_sc_s : ans_to_val (bta_sc_s B) (ans_sc_s A) static_nat (val_sc V)
<- ans_to_val B A S V.

ans_to_val_lam_s : ans_to_val (bta_lam_s B) ans_lam_s static_arrow val_lam.

B.3 Proof of Theorem 3

%h% definition of t3 judgement

t3 : spec WA ->bta ETWP ->btaE’TAP ->ans AP -> conv E E’ -> type.
%%% definition of t3 axiom and rules

% static comnstructs

t3_z_s : t3 (spec_z_s) (bta_z_s) (bta_z_s) (ans_z_s) (conv_refl).

t3_sc_s : t3 (spec_sc_s S) (bta_sc_s B) (bta_sc_s C) (ans_sc_s D) (conv_sc_c E)
<-t3 SBCDE.

t3_prz_s : t3 (spec_prz_s S) (bta_pr_s B) (bta_z_s) (ans_z_s)
(conv_trans (conv_prz) (conv_pr_c E))
<- t3 S B (bta_z_s) D E.

t3_prsc_s : t3 (spec_prsc_s S) (bta_pr_s B) C D
(conv_trans (conv_prsc V)
(conv_pr_c E))
<- ans_to_val C D static_nat V
<- t3 S B (bta_sc_s C) (ans_sc_s D) E.

t3_ifz_s : t3 (spec_ifz_s S2 S1) (bta_if_s B3 B2 B1) C2 D2
(conv_trans E2 (conv_trans (conv_ifz) (conv_if_c1 E1)))
<- t3 S1 B1 (bta_z_s) D1 E1
<- t3 S2 B2 C2 D2 E2.

t3_ifsc_s : t3 (spec_ifsc_s 83 S1) (bta_if_s B3 B2 B1) C3 D3
(conv_trans E3 (conv_trans (conv_ifsc V) (conv_if_c1 E1)))
<- ans_to_val C1 D1 static_nat V
<- t3 S1 B1 (bta_sc_s C1) (ans_sc_s D1) E1
<- t3 S3 B3 C3 D3 E3.
t3_lam_s : t3 (spec_lam_s) (bta_lam_s B) (bta_lam_s B) (ans_lam_s) (conv_refl).

t3_app_s : t3 (spec_app_s S1 S0) (bta_app_s B1 BO) C D

(conv_trans E1 (conv_trans (conv_beta) (conv_app_cO E0)))
<- t3 S0 BO (bta_lam_s CO) DO EO
<- t3 81 (CO _ _ B1) C D Ei.
t3_fix_s : t3 (spec_fix_s S) (bta_fix_s B) C D
(conv_trans E (conv_fix))
<- t3 S (B (fix EO0) (bfix s W0) (bta_fix_s B)) C D E.

% dynamic constructs
t3_z_d : t3 (spec_z_d) (bta_z_d) (bta_z_d) (ans_z_d) (conv_refl).

t3_sc_d : t3 (spec_sc_d S) (bta_sc_d B) (bta_sc_d C) (ans_sc_d D) (conv_sc_c E)
<- t3 S BCDE.

t3_pr_d : t3 (spec_pr_d S) (bta_pr_d B) (bta_pr_d C) (ans_pr_d D) (conv_pr_c E)
<-t3 SBCDE.

t3_if_d : t3 (spec_if_d S3 82 S1) (bta_if_d B3 B2 B1)
(bta_if _d C3 C2 C1) (ams_if_d D3 D2 D1)
(conv_trans (conv_if_c3 E3)
(conv_trans (conv_if_c2 E2)
(conv_if_c1 E1)))
<- t3 S1 B1 C1 D1 E1
<- t3 82 B2 C2 D2 E2
<- t3 S3 B3 C3 D3 E3.

t3_lam_d : t3 (spec_lam_d S) (bta_lam_d B) (bta_lam_d C) (ans_lam_d D)
(conv_lam_c E)
<- {x} {y} {8’: spec y y} {B’ : bta x T3 y dyn} {D’: ans y dyn}
{E’: conv x x}
t3 S’ B> B’ D’ E?
->t3 (SyS’) (BxyB’) (CxyB’) (DyD’) (Ex).

t3_app_d : t3 (spec_app_d S1 S0) (bta_app_d B1 BO)
(bta_app_d C1 C0O) (ans_app_d D1 DO)
(conv_trans (conv_app_cl E1) (conv_app_cO EO0))
<- t3 SO0 BO CO DO EO
<- t3 S1 B1 C1 D1 E1.

t3_fix_d : t3 (spec_fix_d S) (bta_fix_d B) (bta_fix_d C) (ans_fix_d D)
(conv_fix_c E)
<- {x} {y} {S’: spec y y}
{B’ : bta x (arrow T1 T3) y dyn} {D’: ans y dyn} {E’: conv x x}
t3 S’ B’ B’ D’ E’
->t3 (SysS’) (BxyB’) (CxyB’) (DyD’) (Ex).

t3_lift : t3 (spec_lift S) (bta_lift B) (bta_z_d) (ans_z_d) E
<- t3 S B (bta_z_s) D E.

t3_1liftsc : t3 (spec_liftsc 82 S1) (bta_lift B1) (bta_sc_d C2) (ans_sc_d D2)
(conv_trans (conv_sc_c E2) E1)
<- t3 S1 B1 (bta_sc_s C1) D1 E1
<- t3 S2 (bta_lift C1) C2 D2 E2.

